
Abstract
We present flaspland, an Answer Set Programming (ASP) testbed for tackling complex routing and scheduling
problems within the Flatland railway simulation environment. To establish a solid theoretical foundation, we
provide a rigorous formalization of the Flatland environment, detailing its grid-based topology and navigational
dynamics. Building upon this, we propose and analyze three distinct graph-based representations to effectively
encode the underlying constraints. Crucially, we introduce a twofold, path-driven ASP encoding scheme that
decouples spatial routing from temporal scheduling. Our empirical results demonstrate that these specialized rep-
resentations outperform off-the-shelf multi-agent pathfinding approaches. By providing a testbed for real-world
railway scheduling, flaspland facilitates the evaluation of declarative approaches to qualitative and quantitative
temporal constraints.

1. Introduction

The ability of Answer Set Programming (ASP [1]) to easily express reachability makes it a prime candi-
date for addressing routing problems, such as multi-agent path finding [2], phylogenetic inference [3],
and wire routing [4]. However, this advantage is challenged once routing is combined with scheduling to
enforce deadlines and durations, since fine-grained quantitative temporal reasoning cannot be feasibly
represented in ASP.

To address this, we propose railway scheduling as a rich and scalable benchmark for exploring
routing and scheduling problems in ASP and related fields. We specifically advocate for Flatland [5], a
grid-based railway simulation environment developed by major European railway operators, which
enables scalable experimentation with complex railway scenarios. The core task is to solve collision-free
routing and scheduling problems, which are essentially multi-agent pathfinding problems (MAPF; [6])
augmented with quantitative constraints on train arrival and departure times, durations, energy, and
speed.

Figure 1: Visualization of a simple
Flatland scenario with one
train station and one train

The railway network is modeled as a set of connected cells
using matrix notation, as shown in Figure 1. The traversabil-
ity of a cell is defined by its specific track type (see Figure 2),
which dictates valid transitions and may be oriented in 90-
degree increments. The overall objective is to navigate all
trains to their destinations, respecting strict time windows
and preventing collisions. Navigation is performed via dis-
crete actions, whose executability is constrained by the train’s
current direction and the underlying track configuration.

Beyond quantitative scheduling constraints, such railway
networks present structural challenges that differentiate it
from standard MAPF: (i) Track types strictly limit valid cell
transitions, requiring precise modeling of graph connectivity. (ii) A train’s ability to move depends on

its current orientation, adding another layer of complexity to path planning. (iii) The network topology
often features canalized passages with limited switching points, restricting navigational choices and
creating bottlenecks.

In what follows, we provide a formal account of Flatland, detailing its environment, invariants, and
path definitions. Although Flatland’s tasks are intuitively well understood, a rigorous formalization is
necessary to establish the correspondence between railway grids and their graph-based representations,
and ultimately to ensure the correctness and completeness of the ASP encodings. We then propose
alternative graph-based representations of railway networks and establish their correspondence to
the original grid environment. To facilitate empirical evaluation, we introduce the flaspland toolkit,
which bridges ASP and the Flatland environment. Finally, we present a novel two-fold, path-driven
ASP encoding scheme. Our preliminary empirical evaluation demonstrates that these specialized
representations offer significant scalability improvements over off-the-shelf MAPF approaches.

2. Flatland formalization

To begin with, we present some crosscutting concepts. Flatland considers a set 𝐷 = {𝑛, 𝑒, 𝑠, 𝑤} of
cardinal directions and a set 𝑇 of track types.

We define the following operations on directions to compute their opposite and 90-degree rotations:

1. (·) : 𝐷 → 𝐷 where (·) = {𝑛 ↦→ 𝑠, 𝑠 ↦→ 𝑛, 𝑒 ↦→ 𝑤,𝑤 ↦→ 𝑒}
2. (·)𝑟 : 𝐷 → 𝐷 where (·)𝑟 = {𝑛𝑟 ↦→ 𝑒, 𝑒𝑟 ↦→ 𝑠, 𝑠𝑟 ↦→ 𝑤, 𝑤𝑟 ↦→ 𝑛}
3. (·)𝑙 : 𝐷 → 𝐷 where (·)𝑙 = {𝑛𝑙 ↦→ 𝑤, 𝑤𝑙 ↦→ 𝑠, 𝑠𝑙 ↦→ 𝑒, 𝑒𝑙 ↦→ 𝑛}

Clearly, we have 𝑑 = 𝑑, 𝑑2𝑟 = 𝑑2𝑙 = 𝑑, and 𝑑4𝑟 = 𝑑4𝑙 = 𝑑 for 𝑑 ∈ 𝐷.

Figure 2: Basic track types

Flatland’s general track types are shown in Figure 2.1 Type 0 is reserved for empty cells. Specific
track types are derived by rotating these general types in 90-degree increments. For example, the simple
switch at position (2, 1) in Figure 1 corresponds to Type 2 rotated by 90∘, denoted as (2, 90). Each
track type 𝑡 ∈ 𝑇 provides a set of transitions, captured by the relation 𝐻𝑡 ⊆ 𝐷 ×𝐷 satisfying for all
𝑑, 𝑑′ ∈ 𝐷:

1. (𝑑, 𝑑′) ∈ 𝐻𝑡 iff (𝑑′, 𝑑) ∈ 𝐻𝑡

2. (𝑑, 𝑑) ̸∈ 𝐻𝑡

3. |{(𝑑, 𝑑′′) ∈ 𝐻𝑡 | 𝑑′′ ∈ 𝐷}| ≤ 2 and |{(𝑑′′, 𝑑) ∈ 𝐻𝑡 | 𝑑′′ ∈ 𝐷}| ≤ 2

That is, every transition has a complementary one, no transition represents an immediate about-face,
and there may be no more than two transitions for the same incoming and outgoing direction. For
instance, we have 𝐻(2,90) = {(𝑒, 𝑒), (𝑤,𝑤), (𝑒, 𝑛), (𝑠, 𝑤)}.

Notation. The directions 𝐷, track types 𝑇 , and transitions (𝐻𝑡)𝑡∈𝑇 are constant across all Flatland
environments and are therefore assumed fixed in the remainder of this work.

Grids. In Flatland, a railway network is an 𝑛×𝑚 grid of cells 𝐶 . We associate each cell 𝑐 ∈ 𝐶 with a
coordinate position and a track type, via the total functions:

1. 𝑝 : 𝐶 → [0, 𝑛− 1]× [0,𝑚− 1]

1We omit track type 7, the dead end, to align with current versions of Flatland, which no longer produce environments with
this type of track.



2. 𝑡 : 𝐶 → 𝑇

We require 𝑝 to be a bijective function. For the cell with the simple switch,2 we have 𝑝(𝑐2,1) = (2, 1)
and 𝑡(𝑐2,1) = (2, 90). The relation 𝐻𝑡(𝑐) provides the set of transitions for a cell 𝑐 ∈ 𝐶 , and we call 𝑐
trivial whenever 𝐻𝑡(𝑐) = ∅.

The cell adjacent to cell 𝑐 ∈ 𝐶 in the direction 𝑑 ∈ 𝐷 is defined by the partial function

3. 𝑎 : 𝐶 ×𝐷 → 𝐶

such that

𝑝(𝑎(𝑐, 𝑑)) =

⎧⎪⎪⎪⎪⎨⎪⎪⎪⎪⎩
(𝑖− 1, 𝑗) if 𝑝(𝑐) = (𝑖, 𝑗), 𝑑 = 𝑛, and (𝑖, 𝑗), (𝑖− 1, 𝑗) ∈ [0, 𝑛− 1]× [0,𝑚− 1]

(𝑖, 𝑗 + 1) if 𝑝(𝑐) = (𝑖, 𝑗), 𝑑 = 𝑒, and (𝑖, 𝑗), (𝑖, 𝑗 + 1) ∈ [0, 𝑛− 1]× [0,𝑚− 1]

(𝑖+ 1, 𝑗) if 𝑝(𝑐) = (𝑖, 𝑗), 𝑑 = 𝑠, and (𝑖, 𝑗), (𝑖+ 1, 𝑗) ∈ [0, 𝑛− 1]× [0,𝑚− 1]

(𝑖, 𝑗 − 1) if 𝑝(𝑐) = (𝑖, 𝑗), 𝑑 = 𝑤, and (𝑖, 𝑗), (𝑖, 𝑗 − 1) ∈ [0, 𝑛− 1]× [0,𝑚− 1]

Function 𝑎 is partial because of border cells. If 𝑎(𝑐, 𝑑) ̸= 𝑎(𝑐, 𝑑′) then 𝑑 ̸= 𝑑′ for all 𝑐 ∈ 𝐶 and 𝑑, 𝑑′ ∈ 𝐷.
Moreover, we have 𝑎(𝑐, 𝑑) = 𝑐′ iff 𝑎(𝑐′, 𝑑) = 𝑐 for all 𝑐, 𝑐′ ∈ 𝐶 and 𝑑 ∈ 𝐷. For instance, we have
𝑎(𝑐2,1, 𝑤) = 𝑐2,0 and 𝑎(𝑐2,0, 𝑒) = 𝑐2,1, whereas 𝑎(𝑐2,1, 𝑠) is undefined.

A non-trivial cell 𝑐 ∈ 𝐶 is connected to another non-trivial cell 𝑐′ ∈ 𝐶 in direction 𝑑 ∈ 𝐷, written as
𝑐

d→˓ 𝑐′, when the following conditions hold:

1. 𝑎(𝑐, 𝑑) = 𝑐′

2. (𝑑′, 𝑑) ∈ 𝐻𝑡(𝑐) for some 𝑑′ ∈ {𝑑, 𝑑𝑙, 𝑑𝑟}
3. (𝑑, 𝑑′′) ∈ 𝐻𝑡(𝑐′) for some 𝑑′′ ∈ {𝑑, 𝑑𝑙, 𝑑𝑟}

Note that Condition 3 is equivalent to (𝑑′′, 𝑑) ∈ 𝐻𝑡(𝑐′). Moreover, we have 𝑐
d→˓ 𝑐′ iff 𝑐′

d→˓ 𝑐 for
all 𝑐, 𝑐′ ∈ 𝐶 and 𝑑 ∈ 𝐷. Continuing our example, we get 𝑐2,1

w→˓ 𝑐2,0 since 𝑎(𝑐2,1, 𝑤) = 𝑐2,0,
(𝑠, 𝑤) ∈ 𝐻(𝑡(𝑐2,1)), and (𝑤, 𝑛) ∈ 𝐻(𝑡(𝑐2,0)), and hence 𝑐2,0

e→˓ 𝑐2,1, too.
A transit pair (𝑐, 𝑑) combines a non-trivial cell 𝑐 ∈ 𝐶 and direction 𝑑 ∈ 𝐷 such that (𝑑′, 𝑑) ∈ 𝐻𝑡(𝑐) for

some 𝑑′ ∈ 𝐷. The cell 𝑐2,1 gives rise to three transit pairs: (𝑐2,1, 𝑒), (𝑐2,1, 𝑤), and (𝑐2,1, 𝑛), corresponding
to the three possible outgoing directions. We denote the set of all transit pairs for a grid of cells 𝐶 by
𝒯 (𝐶). We require for each transit pair (𝑐, 𝑑) that 𝑐 d→˓ 𝑐′ for some 𝑐′ ∈ 𝐶 . This condition ensures that
each outgoing track leads to an incoming track on the neighboring cell.

We define a transit path as a sequence ((𝑐𝑖, 𝑑𝑖))
𝑛
𝑖=0 of distinct transit pairs with 𝑛 > 0 such that for

all 0 ≤ 𝑖 < 𝑛 :

1. 𝑐𝑖
di→˓ 𝑐𝑖+1

2. (𝑑𝑖, 𝑑𝑖+1) ∈ 𝐻𝑡(𝑐𝑖+1)

Transit paths are non-trivial, as they comprise at least two transit pairs. We call ((𝑐𝑖, 𝑑𝑖))𝑛𝑖=0 a transit
path between (𝑐0, 𝑑0) and (𝑐𝑛, 𝑑𝑛). Condition 2 can be seen as an instantiation of Condition 3 in the
definition of connectedness with the outgoing direction of the subsequent cell. Any portion of a transit
path is itself also a transit path. While a cell may be revisited, each visit must be associated with a
different outgoing direction. Also, all but the last direction capture transitions between cells; the final
one is an arbitrary choice among all outgoing directions of the last cell.

For illustration, consider the following transit path between 𝑐0,1 and 𝑐1,2 in Figure 1:

((𝑐0,1, 𝑠), (𝑐1,1, 𝑠), (𝑐2,1, 𝑤), (𝑐2,0, 𝑛), (𝑐1,0, 𝑒), (𝑐1,1, 𝑒), (𝑐1,2, 𝑒)) (1)

Note that cell 𝑐1,1 is traversed twice, once southwards and once eastwards.
In practice, Flatland environments are always strongly connected. To capture this formally, we

require that for any two cells 𝑐, 𝑐′ ∈ 𝐶 , there is a transit path connecting some transit pairs (𝑐, 𝑑) and
(𝑐′, 𝑑′) with 𝑑, 𝑑′ ∈ 𝐷. Hence, transit paths between cells are not necessarily reversible.
Notation. We associate a Flatland grid with its set 𝐶 of cells and leave functions 𝑝, 𝑡, and 𝑎 implicit.
2The subscript in 𝑐2,1 is just to ease readability.



Trains. Consider a set 𝑍 of trains, each characterized by a start and goal position, an initial orientation,
an earliest departure time, and a latest arrival time, expressed by the functions:

1. 𝑠 : 𝑍 → 𝐶 and 𝑔 : 𝑍 → 𝐶 such that 𝑠(𝑧) ̸= 𝑔(𝑧) for all 𝑧 ∈ 𝑍

2. 𝑜 : 𝑍 → 𝐷 such that (𝑜(𝑧), 𝑑) ∈ 𝐻𝑠(𝑧) for some 𝑑 ∈ 𝐷 and for all 𝑧 ∈ 𝑍

3. 𝛿 : 𝑍 → N and 𝛼 : 𝑍 → N such that 𝛼(𝑧) > 𝛿(𝑧) for all 𝑧 ∈ 𝑍

The provisos stipulate that the starting and goal positions differ for each train, that the initial orientation
of a train is always valid, corresponding to an incoming direction for its starting track, and that time is
discrete, with the arrival time strictly exceeding the departure time.

Notation. We associate functions 𝑠, 𝑔, 𝑜, 𝛿, and 𝛼 with the set 𝑍 of trains and leave them implicit.
A Flatland scenario is a pair (𝐶,𝑍), comprising a set 𝐶 of cells and a set 𝑍 of trains. A train drive is

a transit path that tolerates repeated transit pairs to mimic waiting, meets the train’s initial orientation
and its start and goal position, and respects the earliest departure time. Accordingly, we define a train
drive for a train 𝑧 ∈ 𝑍 and some 𝑚,𝑛 ∈ N in a Flatland scenario (𝐶,𝑍) as a sequence ((𝑐𝑖, 𝑑𝑖))

𝑛
𝑖=𝑚 of

transit pairs such that

1. for all 𝑚 ≤ 𝑖 < 𝑛 either

a) 𝑐𝑖
di→˓ 𝑐𝑖+1 and (𝑑𝑖, 𝑑𝑖+1) ∈ 𝐻𝑡(𝑐𝑖+1), or

b) (𝑐𝑖, 𝑑𝑖) = (𝑐𝑖+1, 𝑑𝑖+1)

2. (𝑐𝑚, 𝑑𝑚) = (𝑠(𝑧), 𝑜(𝑧))

3. 𝑐𝑛 = 𝑔(𝑧) and 𝑐𝑖 ̸= 𝑔(𝑧) for all 𝑚 ≤ 𝑖 < 𝑛

4. 𝑚 ≥ 𝛿(𝑧)

Numbers 𝑚 and 𝑛 reflect the actual departure and arrival times of train 𝑧. The scheduled arrival time,
𝛼(𝑧), is considered separately in delay calculations. Train drives are non-trivial since 𝑠(𝑧) ̸= 𝑔(𝑧).

Actions. Trains have four possible actions:3 move forward (𝑓 ), turn left (𝑙), turn right (𝑟), and wait
(𝑤). A transit pair (𝑐, 𝑑) and one of its transitions (𝑑𝑖, 𝑑) ∈ 𝐻𝑡(𝑐) induce the action 𝑎 where

1. 𝑎 = 𝑓 if
a) 𝐻𝑡(𝑐) ∩ ({𝑑𝑖} ×𝐷) = {(𝑑𝑖, 𝑑)} or
b) 𝐻𝑡(𝑐) ⊃ {(𝑑, 𝑑), (𝑑, 𝑑′)} for 𝑑𝑖 = 𝑑 and some 𝑑′ ∈ {𝑑𝑙, 𝑑𝑟}

2. 𝑎 = 𝑙 if 𝐻𝑡(𝑐) ⊃ {(𝑑𝑟, 𝑑), (𝑑𝑟, 𝑑′)} for 𝑑𝑖 = 𝑑𝑟 and some 𝑑′ ∈ {𝑑, 𝑑𝑟}
3. 𝑎 = 𝑟 if 𝐻𝑡(𝑐) ⊃ {(𝑑𝑙, 𝑑), (𝑑𝑙, 𝑑′)} for 𝑑𝑖 = 𝑑𝑙 and some 𝑑′ ∈ {𝑑, 𝑑𝑙}

𝑖 𝑎𝑖 (𝑐𝑖, 𝑑𝑖)

0 𝑓 (𝑐0,2, 𝑤)
1 𝑓 (𝑐0,1, 𝑠)
2 𝑓 (𝑐1,1, 𝑠)
3 𝑓 (𝑐2,1, 𝑤)
4 𝑓 (𝑐2,0, 𝑛)
5 𝑓 (𝑐1,0, 𝑒)
6 𝑓 (𝑐1,1, 𝑒)
7 𝑤 (𝑐1,2, 𝑒)
8 𝑓 (𝑐1,2, 𝑒)
9 𝑟 (𝑐1,3, 𝑠)
10 𝑓 (𝑐2,3, 𝑤)
11 (𝑐2,2, 𝑤)

Table 1: A plan from 𝑐0,2 to the train sta-
tion in 𝑐2,2 in Figure 1

No navigational choice is required in (1a) since there is
only one alternative when entering from direction 𝑑𝑖. A
second analogous entry point may exist (e.g., track types
6 and 8), but the choice of action 𝑓 remains unambiguous.
The three other cases, (1b), (2) and (3), are analogous in
that they all require choosing 𝑑 as the outgoing direction
from two possible alternatives, using action 𝑓, 𝑙 or 𝑟,
respectively. Transit pairs induce unique actions, either
𝑓 , 𝑙, or 𝑟. Two identical transit pairs form a 𝑤[ait] action.

Given a train drive ((𝑐𝑖, 𝑑𝑖))
𝑛
𝑖=𝑚 for train 𝑧 ∈ 𝑍 in

a Flatland scenario (𝐶,𝑍), we define a plan for train 𝑧
in (𝐶,𝑍) as a sequence (𝑎𝑖)

𝑛−1
𝑖=𝑚 of actions such that for

𝑚 ≤ 𝑖 < 𝑛

1. 𝑎𝑖 = 𝑤 if (𝑐𝑖, 𝑑𝑖) = (𝑐𝑖+1, 𝑑𝑖+1) and
2. 𝑎𝑖 is induced by (𝑐𝑖, 𝑑𝑖) otherwise.

3Flatland provides a fifth action, do nothing, which we consider superfluous and therefore omit from this formalization.



A sample plan for a train 𝑧 is given in Table 1, depicting its journey from the starting cell 𝑠(𝑧) = 𝑐0,2
to the destination station at 𝑔(𝑧) = 𝑐2,2 (as shown in Figure 1). This plan is based on the initial
conditions 𝑜(𝑧) = 𝑛 and 𝛿(𝑧) = 0. The plan requires only two navigational decisions, at steps 6 and 9,
with all other movements being deterministic 𝑓 actions, except for the wait at step 7. The decision
at step 6 is made at cell 𝑐1,1, where the train has two possible transits: one continuing east, via (𝑒, 𝑒),
and another turning south, via (𝑒, 𝑠), both of which are members of 𝐻𝑡(𝑐1,1). The choice of outbound
direction 𝑒 is reflected by the inclusion of (𝑐1,1, 𝑒) in the train drive. Similarly, when transiting 𝑐1,3
eastwards at step 9, the train may go north or south, since both (𝑒, 𝑛) and (𝑒, 𝑠) belong to 𝐻𝑡(𝑐1,3). The
transit pair (𝑐1,3, 𝑠) reflects the southbound trajectory of the train.

Collisions. In Flatland, trains appear upon departure and disappear once they reach their destination.
Two distinct drives

(︀
(𝑐𝑖, 𝑑𝑖)

)︀𝑛
𝑖=𝑚

and
(︀
(𝑐′𝑖, 𝑑

′
𝑖)
)︀𝑛′

𝑖=𝑚′ have

1. a vertex conflict if 𝑐𝑖 = 𝑐′𝑖 for some 𝑖 ∈ [max(𝑚,𝑚′),min(𝑛, 𝑛′)]

2. an edge conflict if 𝑐𝑖+1 = 𝑐′𝑖 and 𝑐𝑖 = 𝑐′𝑖+1 for some 𝑖, 𝑖+ 1 ∈ [max(𝑚,𝑚′),min(𝑛, 𝑛′)]

Solution. A solution to a Flatland scenario (𝐶,𝑍) is a family (𝜋𝑧)𝑧∈𝑍 of plans 𝜋𝑧 for each train 𝑧 ∈ 𝑍
such that all drives underlying (𝜋𝑧)𝑧∈𝑍 are pairwisely vertex and edge conflict-free

The delay of a plan 𝜋𝑧 for a train 𝑧 ∈ 𝑍 is defined as delay(𝜋𝑧) = max(0, 𝑛− 𝛼(𝑧)) where 𝑛 is the
actual arrival time of train 𝑧 with plan 𝜋𝑧 A delay optimal solution to a Flatland scenario (𝐶,𝑍) is a
solution (𝜋𝑧)𝑧∈𝑍 to (𝐶,𝑍) with a minimum value of

∑︀
𝑧∈𝑍 delay(𝜋𝑧)

3. Graph representations

We now consider alternative graph representations for capturing railway networks in Flatland. We
must concern ourselves with the behavior of movement around switches. The adjacency of two cells is a
necessary condition for transition, but not a sufficient one. Furthermore, the direction of the train must
be taken into consideration. For instance, cell 𝑐2,1 has three neighboring cells in Figure 1. However,
only a train facing east can go from 𝑐2,1 to its adjacent cells 𝑐1,1 or 𝑐2,2, respectively, and its neighboring
cell 𝑐2,0 can only be reached by trains facing south or west. In what follows, we consider translations of
grid environments to alternative graph structures which account for such additional qualities.

3.1. Directed graph with edge transition functions

Given a grid 𝐶 , we define the graph 𝒜(𝐶) = (𝑉,𝐸) as

1. 𝐸 =
⋃︀

𝑐∈𝐶,𝑑∈𝐷 {(𝑐, 𝑐′) | 𝑐 d→˓ 𝑐′}
2. 𝑉 = {𝑐, 𝑐′ | (𝑐, 𝑐′) ∈ 𝐸}

The directed graph 𝒜(𝐶) contains no trivial cells from 𝐶 . To accommodate the additional structure
needed to capture transitions, we associate with each graph 𝒜(𝐶) = (𝑉,𝐸) two partial functions on
its set of edges:

1. 𝑑 : 𝐸 → 𝐷 where 𝑑((𝑐, 𝑐′)) = 𝑑 if 𝑐 d→˓ 𝑐′ for (𝑐, 𝑐′) ∈ 𝐸

2. 𝑝 : 𝐸 → 2𝐷 where 𝑝((𝑐, 𝑐′)) = {𝑑′′ | (𝑑′′, 𝑑) ∈ 𝐻𝑡(𝑐)} if 𝑐 d→˓ 𝑐′ for (𝑐, 𝑐′) ∈ 𝐸.

Function 𝑑 reflects the direction of a directed edge, while 𝑝 gives the necessary arriving directions when
leaving a node.

Figure 3(a) shows the result of translating our example grid from Figure 2. First of all, note that the
trivial cell 𝑐0,0 was dropped from the graph. To illustrate the purpose of both functions, consider node
𝑐2,1 in Figure 3(a) along with function 𝑑 and 𝑝 applied to its incoming and outgoing edges, respectively.

𝑑((𝑐2,0, 𝑐2,1)) = 𝑒 𝑝((𝑐2,1, 𝑐2,0)) = {𝑠, 𝑤} (2)



c2,0 c2,1 c2,2 c2,3

c1,0 c1,1 c1,2 c1,3

c0,1 c0,2 c0,3

Figure 3: Representation of our example as a directed graph with (a) edge transition functions and (b) sub-nodes.

𝑑((𝑐1,1, 𝑐2,1)) = 𝑠 𝑝((𝑐2,1, 𝑐2,2)) = {𝑒} (3)

𝑑((𝑐2,2, 𝑐2,1)) = 𝑤 𝑝((𝑐2,1, 𝑐1,1)) = {𝑒} (4)

The interplay of both functions determines whether an incoming and outgoing edge allow for a transition
over a node at hand. More precisely, this is the case for a node 𝑐, whenever 𝑑((𝑐′, 𝑐)) ∈ 𝑝((𝑐, 𝑐′′)) for
some connected nodes 𝑐′, 𝑐′′. Four combinations of the mappings above satisfy this condition for 𝑐2,1,
which reflect the four possible transitions over 𝑐2,1.

Given a grid 𝐶 , we define a path in 𝒜(𝐶) as a sequence (𝑐𝑖)
𝑛
𝑖=0 of vertices with 𝑛 > 0 such that

1. (𝑐𝑖, 𝑐𝑖+1) ∈ 𝐸 for all 0 ≤ 𝑖 < 𝑛

2. 𝑑((𝑐𝑖−1, 𝑐𝑖)) ∈ 𝑝((𝑐𝑖, 𝑐𝑖+1)) for all 0 < 𝑖 < 𝑛

3. if 𝑐𝑖 = 𝑐𝑗 then 𝑐𝑖+1 ̸= 𝑐𝑗+1 for all 0 ≤ 𝑖 < 𝑗 < 𝑛

We call (𝑐𝑖)𝑛𝑖=0 a path between 𝑐0 and 𝑐𝑛 in𝒜(𝐶). For example, the transit path ((𝑐1,1, 𝑠), (𝑐2,1, 𝑤), (𝑐2,0, 𝑛))
between 𝑐1,1 and 𝑐2,0 in our grid 𝐶 in Figure 1 corresponds to the path (𝑐1,1, 𝑐2,1, 𝑐2,0) in 𝒜(𝐶) and
satisfies 𝑑((𝑐1,1, 𝑐2,1)) ∈ 𝑝((𝑐2,1, 𝑐2,0)).

The following proposition shows that our grid representation preserves all paths in the original grid.

Proposition 1. Let 𝐶 be a grid.

1. If ((𝑐𝑖, 𝑑𝑖))𝑛𝑖=0 is a transit path in grid 𝐶 , then (𝑐𝑖)
𝑛
𝑖=0 is a path in the graph 𝒜(𝐶).

2. If (𝑐𝑖)𝑛𝑖=0 is a path in the graph 𝒜(𝐶), then ((𝑐𝑖, 𝑑𝑖))
𝑛
𝑖=0 is a transit path in grid 𝐶 ,

where 𝑑𝑖 = 𝑑((𝑐𝑖, 𝑐𝑖+1)) for 0 ≤ 𝑖 < 𝑛.

3.2. Directed graph with sub-nodes

Given a grid 𝐶 , we define the graph ℬ(𝐶) = (𝑉,𝐸) as

1. 𝑉 = 𝒯 (𝐶)

2. 𝐸 = {((𝑐, 𝑑), (𝑐′, 𝑑′)) ∈ 𝑉 × 𝑉 | 𝑐 d→˓ 𝑐′, (𝑑, 𝑑′) ∈ 𝐻𝑡(𝑐′)}

The directed graph ℬ(𝐶) connects all transit pairs of a grid. Fragmenting cells into transit pairs and
relating them in terms of their connectivity allows us to account for the transitions over the various
track types.

The result of translating our Flatland grid from Figure 2 is given in Figure 3(b). The cell 𝑐2,1 is now
decomposed into three nodes, (𝑐2,1, 𝑒), (𝑐2,1, 𝑤), and (𝑐2,1, 𝑛), corresponding to its three transit pairs.
The dashed circles in Figure 3(b) are only meant to indicate nodes stemming from the same original
cell and have no structural meaning. Intuitively, the above edge transition functions are compiled into
the refined graph structure. Each node (or transit pair) represents one way of exiting a node. The four
possible passages over 𝑐2,1 are reflected by the four edges entering the three nodes derived from 𝑐2,1.

Given a grid 𝐶 , we define a path in ℬ(𝐶) as a sequence ((𝑐𝑖, 𝑑𝑖))𝑛𝑖=0 of transit pairs with 𝑛 > 0 such
that

1. ((𝑐𝑖, 𝑑𝑖), (𝑐𝑖+1, 𝑑𝑖+1)) ∈ 𝐸 for all 0 ≤ 𝑖 < 𝑛



2. if (𝑐𝑖, 𝑑𝑖) = (𝑐𝑗 , 𝑑𝑗), then 𝑖 = 𝑗 for all 0 ≤ 𝑖 < 𝑗 < 𝑛

We call ((𝑐𝑖, 𝑑𝑖))𝑛𝑖=0 a path between (𝑐0, 𝑑0) and (𝑐𝑛, 𝑑𝑛) inℬ(𝐶). In fact, the transit path ((𝑐1,1, 𝑠), (𝑐2,1, 𝑤), (𝑐2,0, 𝑛))
between 𝑐1,1 and 𝑐2,0 in our grid 𝐶 in Figure 1 is also a path in ℬ(𝐶).

Proposition 2. Let 𝐶 be a grid.

1. If ((𝑐𝑖, 𝑑𝑖))𝑛𝑖=0 is a transit path in grid 𝐶 , then ((𝑐𝑖, 𝑑𝑖))
𝑛
𝑖=0 is a path in the graph ℬ(𝐶).

2. If ((𝑐𝑖, 𝑑𝑖))𝑛𝑖=0 is a path in the graph ℬ(𝐶), then ((𝑐𝑖, 𝑑𝑖))
𝑛
𝑖=0 is a transit path in grid 𝐶 .

3.3. Directed graph with hyper-edges

Given a grid 𝐶 , we define the graph ℋ(𝐶) = (𝑉,𝐸) as

1. 𝐸 =
⋃︀

𝑐∈𝐶,𝑑∈𝐷 {(𝑐, 𝑐′, 𝑐′′) | 𝑐 d→˓ 𝑐′, 𝑐′
d ′
→˓ 𝑐′′, (𝑑, 𝑑′) ∈ 𝐻𝑡(𝑐′)}

2. 𝑉 = {𝑐, 𝑐′, 𝑐′′ | (𝑐, 𝑐′, 𝑐′′) ∈ 𝐸}

ℋ(𝐶) is a graph using directed hyper-edges among the non-trivial cells of the original grid 𝐶 . Each
hyper-edge captures a valid passage over a node by connecting it to its predecessor and successor in
this transition.

Consider the graph ℋ(𝐶) obtained from our Flatland grid in Figure 2. We get four hyper-edges
centered upon node 𝑐2,1, namely,

(𝑐2,0, 𝑐2,1, 𝑐2,2) (𝑐2,0, 𝑐2,1, 𝑐1,1) (5)

(𝑐1,1, 𝑐2,1, 𝑐2,0) (𝑐2,2, 𝑐2,1, 𝑐2,0) (6)

These hyper-edges correspond to the four passages over 𝑐2,1 discussed above.
Given a grid 𝐶 , we define a path in ℋ(𝐶) as a sequence (𝑐𝑖)

𝑛
𝑖=0 of vertices with 𝑛 > 0 such that

1. (𝑐𝑖−1, 𝑐𝑖, 𝑐𝑖+1) ∈ 𝐸 for all 0 < 𝑖 < 𝑛

2. (𝑐, 𝑐0, 𝑐1) ∈ 𝐸 and (𝑐𝑛−1, 𝑐𝑛, 𝑐
′) ∈ 𝐸 for some 𝑐, 𝑐′ ∈ 𝑉

3. if 𝑐𝑖 = 𝑐𝑗 then 𝑐𝑖+1 ̸= 𝑐𝑗+1 for all 0 ≤ 𝑖 < 𝑗 < 𝑛

We call (𝑐𝑖)𝑛𝑖=0 a path between 𝑐0 and 𝑐𝑛 in ℋ(𝐶). Condition 2 ensures that the constraints imposed by
hyper-edges are satisfied at the endpoints of each path, and for paths of length two.

For example, the transit path ((𝑐1,1, 𝑠), (𝑐2,1, 𝑤), (𝑐2,0, 𝑛)) between 𝑐1,1 and 𝑐2,0 in our grid 𝐶 in
Figure 1 corresponds to the path (𝑐1,1, 𝑐2,1, 𝑐2,0) in ℋ(𝐶). For ℋ(𝐶) = (𝑉,𝐸), it satisfies the following
conditions:

(𝑐0,1, 𝑐1,1, 𝑐2,1) ∈ 𝐸 (7)

(𝑐1,1, 𝑐2,1, 𝑐2,0) ∈ 𝐸 (8)

(𝑐2,1, 𝑐2,0, 𝑐1,0) ∈ 𝐸 (9)

Proposition 3. Let 𝐶 be a grid.

1. If ((𝑐𝑖, 𝑑𝑖))𝑛𝑖=0 is a transit path in grid 𝐶 , then (𝑐𝑖)
𝑛
𝑖=0 is a path in the graph ℋ(𝐶).

2. If (𝑐𝑖)𝑛𝑖=0 is a path in the graph ℋ(𝐶), then ((𝑐𝑖, 𝑑𝑖))
𝑛
𝑖=0 is a transit path in grid 𝐶 ,

where 𝑐𝑖
di→˓ 𝑐𝑖+1 for 0 ≤ 𝑖 < 𝑛.



4. Flaspland : Flatland to ASP and back

Flaspland is an open-source toolkit that bridges the gap between ASP and the Flatland ecosystem . The
toolkit provides a bidirectional translation layer between instance generation and solving. Flatland
environments are converted into instance files that adhere to a consistent ASP fact format, while
answer sets produced by custom encodings are mapped back to Flatland actions for execution and
visualization. Rather than requiring users to construct instances manually, flaspland integrates directly
with the existing environment generator in Flatland and supports batch generation of environments to
streamline experimental workflows at scale. Additionally, the toolkit serves as a repository housing
Flatland benchmark instances, providing a unified point of access for reproducible evaluation. These
comprise the 150 environments used in the final round of the Flatland 3 competition. The environments
range in size, number of agents and cities, and include trains that malfunction and move at various
speeds. The flaspland toolkit is freely and publicly available at [7].

5. Encodings

We now present the ASP encodings for addressing routing and scheduling in Flatland, starting with the
representation of the environment’s basic topological and navigational concepts. These encodings are
available at [8]. The fact format of the basic concepts regarding directions is given in Listing 1.

1 dir(n). dir(s). dir(w). dir(e).
2 inverse(n,s). inverse(s,n). inverse(w,e). inverse(e,w).
3 left(n,w). left(s,e). left(w,s). left(e,n).
4 right(n,e). right(s,w). right(w,n). right(e,s).
5 offset(n,-1,0). offset(s,1,0). offset(w,0,-1). offset(e,0,1).

Listing 1: Representation of directions (n,s,w,o)

Predicate dir/1 gives the cardinal directions, inverse/2 provides complementary directions, left/2
and right/2 capture 90-degree rotations, and offset/3 gives a coordinate offset when moving in a
certain direction.

In Flatland, each track type has a numeric value that is based on its possible transitions;4 it is
represented via predicate transition/2, one instance per transition. Listing 2 contains all facts
reflecting the four transitions of the simple switch at position (2, 1) in Figure 1.

42 transition(3089,(s,w))
43 transition(3089,(w,w))
44 transition(3089,(e,e))
45 transition(3089,(e,n))

Listing 2: Selected lines of track type representations

The actual network is represented by atoms of predicate cell/2, where the first argument reflects
the coordinates of each cell and the second gives its type (cf. Listing 4). The first argument serves as
both the identifier and the value of function 𝑝 for the cell at hand. The second argument gives the
cell’s track type; its transitions are obtained by combining predicates cell/2 and transition/2.
The adjacency function 𝑎 and the connectivity relation →˓ are defined in Listing 3, using predicates
adjacent/3 and connected/3, respectively.

1 adjacent((X,Y),D,(X’,Y’)) :- cell((X,Y),_), cell((X’,Y’),_),
2 offset(D,DX,DY), X’=X+DX, Y’=Y+DY.

4 connected((X,Y),D,(X’,Y’)) :- cell((X,Y),T), cell((X’,Y’),T’),

4For example, the switch track at coordinate (2, 1) in Figure 1 has the value 3089, based on its 16-digit binary representation
0000 1100 0001 0001 of the transitions in 𝐻(2,1). Each group of four digits represents transitions when an agent is
facing north, east, south, or west, respectively. Within each group, the four digits represent transitions out of a cell in the
directions north, east, south, or west, respectively. The digit 1 indicates the transition is permissible.



5 adjacent((X,Y),D,(X’,Y’)),
6 transition(T,(D’,D)), transition(T’,(D,D’’)).

Listing 3: Representation of grid functions

In addition to representing the grid with the cell/2 predicate, our instance format uses start/4
and end/3 to fix the start and end positions of trains. The first two arguments of these predicates
specify the train identifier 𝑧 and the starting 𝑠(𝑧) or ending 𝑔(𝑧) position, respectively. The third and
fourth arguments of start/4 indicate the earliest departure time 𝛿(𝑧) and initial orientation 𝑜(𝑧),
while the third argument of end/3 specifies the latest arrival time 𝛼(𝑧). An example is given in Line 1
and 2 in Listing 4.

1 start(0,(0,2),0,w).
2 end(0,(2,2),12).

4 cell((0,0), 0). cell((0,1),16386). cell((0,2),1025). cell((0,3),4608).
5 cell((1,0),16386). cell((1,1),38433). cell((1,2),1025). cell((1,3),6672).
6 cell((2,0), 72). cell((2,1), 3089). cell((2,2),1025). cell((2,3),2064).

Listing 4: The fact representation of our running example (Figure 1)

The fact representation introduced above, generated directly from Flatland scenarios by our flaspland
framework, constitutes the basic input for our ASP encodings. Building upon this basis, we investigate
how the three graph-based representations defined in Section 3 can be utilized to effectively model
train dynamics and interactions.

Our encoding design is driven by two main objectives. First, we aim to evaluate the computational
trade-offs of the proposed topological representations, specifically regarding how the choice of structural
abstraction influences ASP solving performance. Second, our encodings must robustly address the
spatio-temporal complexities inherent to train scheduling. Foremost is the need to model waiting
actions for collision avoidance, which inherently decouples spatial transitions from strict temporal
succession. Further complexities stem from the asynchronous life-cycle of trains, such as their variable
appearance, delayed departures, and disappearance upon arrival. Moreover, a train’s initial orientation
can typically only be verified upon its actual departure.

We address these spatio-temporal complexities by decomposing the synthesis of train drives into
two distinct aspects: spatial routing and temporal scheduling. The routing component establishes the
backbone of a train’s journey by identifying a valid path between its start and destination. Crucially,
while these paths may revisit nodes, they prohibit consecutive duplicate nodes. This abstraction of
temporal waiting ensures that spatial transitions can be directly verified against the underlying track
topology. The scheduling component then refines these spatial paths into fully realized train drives by
inserting appropriate dwelling times to enforce collision avoidance (Section 5.5).

In the following, we provide a specialized path encoding for each graph representation and subse-
quently integrate them into a unified drive encoding in Section 5.4.

5.1. Encodings based on directed graphs with edge transition functions

This section introduces encodings based on a basic graph representation augmented with edge functions,
as formalized in Section 3.1. The encoding in Listing 5 translates a Flatland grid into this structure.

1 edge(C,C’) :- connected(C,_,C’).
2 node(C;C’) :- edge(C,C’).

4 d(C,C’,D) :- connected(C,D,C’).
5 p(C,C’,D) :- connected(C,D’,C’), cell(C,T), transition(T,(D,D’)).

Listing 5: Encoding for directed graphs with edge functions (cf. Section 3.1).

The graph topology is captured by the predicates node/1 and edge/2, while d/3 and p/3 encode the
functions 𝑑 and 𝑝, respectively.



The encoding in Listing 6 identifies spatial paths of length up to h between each train’s starting and
goal positions.
1 time(0..h).

3 path(Z,C,0; Z,C’,1) :- start(Z,C,_,D), adjacent(C,D,C’).
4 { path(Z,C’,T+1): edge(C,C’) } = 1 :- path(Z,C,T), time(T+1), not eop(Z,T).

6 eop(Z,T) :- end(Z,C,_), path(Z,C,T).
7 :- start(Z,_,_,_), not eop(Z,_).

9 :- path(Z,C,T), path(Z,C’,T+1), path(Z,C’’,T+2), d(C,C’,D), not p(C’,C’’,D).

Listing 6: Pathfinding with edge-functions

Line 1 defines the available time steps, while Line 3 initializes each train at its designated starting position
and orientation. The path is iteratively extended (Line 4) until the destination is reached. Termination
is signaled by the eop(Z,T) atom (Line 6),5 indicating that train Z has arrived at its destination at time
step T. The integrity constraint in Line 7 ensures that every train eventually reaches its goal. Crucially,
Line 9 enforces transition validity by applying the edge functions 𝑑 and 𝑝 to consecutive nodes in the
path, verifying that the incoming and outcoming edge form a valid transition.

5.2. Encodings based on directed graphs with sub-nodes

This section presents encodings leveraging the augmented node structure defined in Section 3.2. The
translation from the standard Flatland format into this representation is handled by the encoding in
Listing 7.

1 node((X,Y,D)) :- cell((X,Y),T), transition(T, (_,D)).

3 edge((X,Y,D), (X’,Y’,D’)) :- node((X,Y,D)), node((X’,Y’,D’)),
4 connected((X,Y),D,(X’,Y’)), cell((X’,Y’), T’), transition(T’, (D,D’)).

Listing 7: Encoding for directed graphs with sub-nodes (cf. Section 3.2).

In this model, each node corresponds to a transit pair (a cell and an outgoing direction), thereby encoding
valid transitions directly into the graph’s edge construction.

Listing 8 adapts the pathfinding logic to this refined structure.
1 time(0..h).

3 path’(Z,(X,Y,D),0) :- start(Z,(X,Y),_,D).
4 { path’(Z,C’,T+1) : edge(C,C’) } = 1 :- path’(Z,C,T), time(T+1), not eop(Z,T).

6 eop(Z,T) :- end(Z,(X,Y),_), path(Z,(X,Y),T).
7 :- start(Z,_,_,_), not eop(Z,_).

9 path(Z,(X,Y),T) :- path’(Z,(X,Y,D),T).

Listing 8: Pathfinding with sub-nodes

While the rules in Lines 1–7 mirror the logic of the edge-function-based encoding, they operate on transit
pairs rather than simple cells. To provide a uniform interface for subsequent scheduling, the internal
predicate path’/3 is projected back onto the cell-based path/3 in Line 9. In this representation,
transition validity is guaranteed by construction and requires no explicit verification during path
generation.

5.3. Encodings based on directed graphs with hyper-edges

This section introduces encodings utilizing the hypergraph structure defined in Section 3.3. Listing 9
provides the translation into this representation.

5eop stands for “end of path”.



1 hedge(C,C’,C’’) :- connected(C,D,C’), connected(C’,D’,C’’),
2 cell(C’,T), transition(T,(D,D’)).

4 node(C;C’;C’’) :- hedge(C,C’,C’’).

Listing 9: Encoding for directed graphs with hyper-edges (cf. Section 3.3).

While node/1 continues to designate cell-based nodes, the valid transitions are modeled via the ternary
predicate hedge/3. This approach naturally captures the dependencies between a node’s entry and
exit points.

Listing 10 details the corresponding pathfinding encoding.
1 time(0..h).

3 path(Z,C,0; Z,C’,1) :- start(Z,C,_,D), adjacent(C,D,C’).
4 { path(Z,C’’,T+1) : hedge(C,C’,C’’) } = 1 :- path(Z,C,T-1), path(Z,C’,T), time(T+1), not eop(Z,T).

6 eop(Z,T) :- end(Z,C,_), path(Z,C,T).
7 :- start(Z,_,_,_), not eop(Z,_).

Listing 10: Pathfinding with hyper-edges

The path generation rules (Lines 1–7) leverage the hyper-edges to ensure that every step in the path is
valid according to the underlying track topology. This representation is arguably the most direct, as the
hyper-edges explicitly encapsulate the ternary nature of cell transitions without requiring auxiliary
functions or node fragmentation.

5.4. Encodings of train drives

To transform spatial paths into fully realized train drives, we must account for potential dwelling
times at each node, which are essential for collision avoidance. The total dwelling time for a train is
constrained by the interval between its earliest departure and its actual arrival at the destination. This
constraint is captured by the predicate delay_budget/2 (Line 1). The cumulative delay at each node
along the path is then defined in Lines 3 to 5. Once a specific delay is attributed to a node in Line 3,
Line 4 ensures its monotonic accumulation, while Line 5 terminates this upon reaching the goal. Note
that we use the index −1 as a convention to represent the initial delay in departing compared to the
earliest possible start time. Finally, the resulting drive is assembled in Line 7 by mapping the spatial
path coordinates to their respective temporal steps based on these accumulated delays.
1 delay_budget(Z,h-L-S) :- start(Z,_,S,_), eop(Z,L).

3 { delay(Z,T-1,0..B) } = 1 :- path(Z,_,T), delay_budget(Z,B).
4 :- delay(Z,T,W), delay(Z,T+1,W’), W’ < W.
5 delay(Z,T,D) :- eop(Z,T), delay(Z,T-1,D).

7 drive(Z,C,T+S+W..T+S+W’) :- path(Z,C,T), start(Z,_,S,_), delay(Z,T-1,W), delay(Z,T,W’).

9 #show drive/3.

Listing 11: Assembling train drives from paths

5.5. Encodings of collisions, plans, and delay optimality

Beyond individual train dynamics, the feasibility of a solution hinges on the absence of inter-train
interference. To this end, the vertex and edge conflict conditions established in Section 2 are implemented
in Listing 12. These constraints ensure that no two trains occupy the same cell simultaneously and
prevent head-on collisions by forbidding trains from traversing the same edge in opposite directions at
the same time step.
1 :- drive(Z,C,T), drive(Z’,C,T), Z < Z’.
2 :- drive(Z,C,T), drive(Z,C’T+1), drive(Z’,C’,T), drive(Z’,C,T+1), Z < Z’.

Listing 12: Vertex and edge conflicts



MAPF Edge-functions Sub-nodes Hyper-edges

|𝑇 | |𝐼| Grnd Global TO Rules Vars Cons Grnd Global TO Rules Vars Cons Grnd Global TO Rules Vars Cons Grnd Global TO Rules Vars Cons

2 70 3.1 8.9 0 801516 298688 767748 38.9 86.4 0 10424661 4250093 21649502 19.7 20.2 0 4545471 2004087 10059450 36.4 50.1 0 9371445 3932519 19886951
3 70 4.5 81.1 0 1189088 427171 1130507 81.8 201.3 0 21709988 7309423 41164599 45.2 47.4 0 10757686 3413388 20093834 79.6 119.2 0 19907260 6851270 38352250
4 70 6.0 483.9 8 1600794 547378 1484530 139.2 585.6 8 35679753 10507063 62477860 82.5 133.5 1 19682594 4964144 33211557 140.4 333.8 5 32704159 9805305 58390218
5 70 7.5 1069.2 31 2011070 644632 1774885 190.6 468.8 4 25063841 12968607 40274980 129.5 196.2 2 29441568 6306472 45867073 190.3 273.1 3 26650172 12167195 43820106
6 70 9.3 1378.1 56 2462364 755469 2118716 246.4 580.7 10 28215589 15212880 43269829 183.2 306.3 6 33287181 7631051 48155418 243.0 296.2 1 28711553 14083026 44824302
7 10 10.4 1500.0 10 2486713 845793 2223014 260.4 286.1 0 8685563 7852626 11475302 235.0 462.4 1 38357113 8967436 52883832 236.0 252.4 0 7956530 7324500 10824910

Table 2
Average experimental results comparing various drive encoding strategies.

Finally, we address the optimization of the overall schedule. While the pathfinding component
identifies valid spatial routes, the scheduling component must minimize the total deviation from the
target arrival times. This is achieved by minimizing the sum of delays across all trains, effectively
penalizing any dwelling time that extends a train’s journey beyond its scheduled window. By combining
these constraints, the resulting ASP encoding provides a complete framework for synthesizing delay-
optimal solutions within the Flatland environment.
1 #minimize{ D : delay(Z,T,D), eop(Z,T) }.

Listing 13: Delay minimization

Finally, we note that the concrete sequences of actions, or plans, are implicitly defined by the underlying
train drives. Within the flaspland framework, these plans are primarily utilized for visualization
purposes and are therefore omitted here for conciseness.

6. Experiments
We evaluate the three proposed topological representations against a baseline encoding on 360 bench-
mark instances to assess their computational performance and scalability. The baseline is an adaptation
of a mature, off-the-shelf MAPF encoding [9], augmented with the necessary constraints to handle Flatland’s
directional dynamics. Our benchmark set is derived from the first ten base environments described in Section 4.
Each environment consists of a 30 × 30 grid populated with seven trains. To investigate how our encodings
scale with agent density, we systematically varied the number of active trains while maintaining a fixed track
topology. Specifically, for each of the ten base environments and for each subset size 𝑘 ∈ {2, . . . , 6}, we randomly
sampled seven train combinations. For 𝑘 = 7, the unique full instance was included. This sampling strategy
yields 10× (5× 7+1) = 360 benchmark instances in total. For each instance, we set the maximum time horizon
to the latest arrival time among the participating trains, ensuring a feasible temporal window for all agents. We
ran all instances on a cluster with Intel Xeon E5-2650v4@2.9GHz CPUs with 64GB of memory running Debian
Linux 10. We used clingo 5.8.0 and imposed a timeout of 1500 seconds and a memory limit of 15GB per run.

Table 2 summarizes our results, comparing the four configurations MAPF, Edge-functions, Sub-nodes, and
Hyper-edges, across scenarios with an increasing number of trains (|T|, from two to seven). The off-the-shelf
MAPF configuration scales poorly as the number of trains increases. While it produces the smallest problem sizes
(featuring fewer rules (Rules), variables (Vars), and constraints (Cons))6 and boasts the fastest grounding times
(Grnd, averaging 6.8s), it struggles to solve complex instances. It reaches a peak of 56 timeouts (TO) for six trains
and hits the 1500-second time limit for all ten instances (|I|) with seven trains. In contrast, the three path-driven
variants generate an order of magnitude more rules and variables but achieve significantly better overall success
rates and lower total solving times (Global) for larger train counts.

Among the proposed graph representations, Sub-nodes and Hyper-edges significantly outperform the Edge-
functions variant in both global time and reliability. Edge-functions consistently records the highest global times
(averaging 368.16s) and the most timeouts (averaging 3.67), making it the least robust of the three. Interestingly,
Sub-nodes generates the largest instantiations (averaging 22.68M rules) compared to Edge-functions (21.63M) and
Hyper-edges (20.88M), yet it achieves the fastest grounding times, averaging 115.85s, roughly 25% faster than the
other two variants. Regarding overall scalability, Sub-nodes provides the most consistent performance for smaller
instances, starting with a minimal global time of 20.17s for two trains. However, it exhibits a steeper increase in
global time as the system scales, reaching 462.40s for seven trains. Conversely, Hyper-edges demonstrates better
scalability under heavy loads. While starting slightly slower (50.06s for two trains), it maintains a much more
stable trajectory, ending at 252.36s for seven trains and yielding the lowest average number of timeouts (1.5).

6Specifically, Rules denotes the number of ground rules generated by the grounder, while Vars and Cons represent
the number of Boolean variables and constraints, respectively, internally translated and processed by the solver.



7. Conclusion
While Flatland was originally conceived as a benchmark for reinforcement learning [10], the success of symbolic
MAPF techniques in the 2020 Flatland competition highlighted the strengths of combinatorial approaches in
this domain [11]. Building upon the successful application of ASP to real-world railway scheduling [12], we
introduced flaspland as a dedicated testbed for tackling these challenges within a declarative framework.

This preliminary report provides a rigorous formalization of the Flatland environment, establishing a solid
foundation for verifying the correctness of subsequent ASP encodings. We propose three distinct graph-based
representations to capture the environment’s topological and navigational constraints, and develop a two-phase
ASP scheme that effectively decouples spatial routing from temporal scheduling. Our empirical evaluation
demonstrates that this path-driven representation, particularly those employing sub-nodes and hyper-edges, offer
significant scalability advantages over off-the-shelf MAPF encodings. By bridging the gap between grid-based
simulation and graph-based encodings, flaspland provides a scalable benchmark for evaluating the performance
of ASP in complex, multi-agent railway scenarios.

8. Declaration on Generative AI
The authors have not employed any generative AI tool in the preparation of this work.

